# Borwein Integrals

Sep 23, 2018
Justin Pearson

Source: https://www.futilitycloset.com/2018/02/02/breakdown-2/

## Problem statement

Check out this strange behavior:

In[1]:= $\int_0^\infty \frac{\text{Sin}[x]}{x} \, dx$

Out[1]= $\frac{\pi}{2}$

In[2]:= $\int_0^\infty \frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \, dx$

Out[2]= $\frac{\pi}{2}$

In[3]:= $\int_0^\infty \frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \frac{\text{Sin}[x/5]}{x/5} \, dx$

Out[3]= $\frac{\pi}{2}$

In[4]:= $\int_0^\infty \frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \frac{\text{Sin}[x/5]}{x/5} \frac{\text{Sin}[x/7]}{x/7} \, dx$

Out[4]= $\frac{\pi}{2}$

In[5]:= $\int_0^\infty \frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \frac{\text{Sin}[x/5]}{x/5} \frac{\text{Sin}[x/7]}{x/7} \frac{\text{Sin}[x/9]}{x/9} \, dx$

Out[5]= $\frac{\pi}{2}$

In[6]:= $\int_0^\infty \frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \frac{\text{Sin}[x/5]}{x/5} \frac{\text{Sin}[x/7]}{x/7} \frac{\text{Sin}[x/9]}{x/9} \frac{\text{Sin}[x/11]}{x/11} \, dx$

Out[6]= $\frac{\pi}{2}$

In[7]:= $$\int_0^\infty \frac{\text{Sin}[x]}{x}\frac{\text{Sin}[x/3]}{x/3}\frac{\text{Sin}[x/5]}{x/5}\frac{\text{Sin}[x/7]}{x/7}\frac{\text{Sin}[x/9]}{x/9}\frac{\text{Sin}[x/11]}{x/11}\frac{\text{Sin}[x/13]}{x/13}\,dx$$

Out[7]= $$\frac{\pi}{2}$$

In[8]:= $$\int_0^\infty \frac{\text{Sin}[x]}{x}\frac{\text{Sin}[x/3]}{x/3}\frac{\text{Sin}[x/5]}{x/5}\frac{\text{Sin}[x/7]}{x/7}\frac{\text{Sin}[x/9]}{x/9}\frac{\text{Sin}[x/11]}{x/11}\frac{\text{Sin}[x/13]}{x/13}\frac{\text{Sin}[x/15]}{x/15}\,dx$$

Out[8]= $$\frac{467\,807\,924\,713\,440\,738\,696\,537\,864\,469\,\pi}{935\,615\,849\,440\,640\,907\,310\,521\,750\,000}$$

This is quite close to $\pi/2$:

In[9]:= $N[\%] - \frac{\pi}{2}$

Out[9]= $-2.31006\times10^{-11}$

The next term in the series is a little farther away from $\pi/2$:

In[10]:= $$\int_0^\infty \frac{\text{Sin}[x]}{x}\frac{\text{Sin}[x/3]}{x/3}\frac{\text{Sin}[x/5]}{x/5}\frac{\text{Sin}[x/7]}{x/7}\frac{\text{Sin}[x/9]}{x/9}\frac{\text{Sin}[x/11]}{x/11}\frac{\text{Sin}[x/13]}{x/13}\frac{\text{Sin}[x/15]}{x/15}\frac{\text{Sin}[x/17]}{x/17}\,dx$$

Out[10]= $$\frac{17\,708\,695\,183\,056\,190\,642\,497\,315\,530\,628\,422\,295\,569\,865\,119\,\pi}{35\,417\,390\,788\,301\,195\,294\,898\,352\,987\,527\,510\,935\,040\,000\,000}$$

In[11]:= $N[\%] - \frac{\pi}{2}$

Out[11]= $-1.87245\times10^{-8}$

It only gets worse from there:

In[12]:= $$\int_0^\infty \frac{\text{Sin}[x]}{x}\frac{\text{Sin}[x/3]}{x/3}\frac{\text{Sin}[x/5]}{x/5}\frac{\text{Sin}[x/7]}{x/7}\frac{\text{Sin}[x/9]}{x/9}\frac{\text{Sin}[x/11]}{x/11}\frac{\text{Sin}[x/13]}{x/13}\frac{\text{Sin}[x/15]}{x/15}\frac{\text{Sin}[x/17]}{x/17}\frac{\text{Sin}[x/19]}{x/19}\,dx$$

Out[12]= $$\frac{8\,096\,799\,621\,940\,897\,567\,828\,686\,854\,312\,535\,486\,311\,061\,114\,550\,605\,367\,511\,653\,\pi}{16\,193\,600\,755\,941\,299\,921\,751\,838\,065\,715\,269\,433\,640\,150\,152\,124\,763\,150\,000\,000}$$

In[13]:= $N[\%] - \frac{\pi}{2}$

Out[13]= $-1.46671\times10^{-7}$

What’s going on here?

# Analysis

Each integral is a product of scaled sinc functions, which corresponds to a convolution of scaled rect functions in the Fourier domain.

$\int_0^{\infty} \frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \ldots dx$

$= \frac{1}{2} \int_{-\infty}^{\infty} \frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \ldots dx$ functions are all even

$= \frac{1}{2} \left(\mathcal{F}\left(\frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \ldots\right)\right)(0)$ def of fourier transform

$= \frac{1}{2} \left(\mathcal{F}\left(\frac{\text{Sin}[x]}{x}\right) \star \mathcal{F}\left(\frac{\text{Sin}[x/3]}{x/3}\right) \star \ldots\right)(0)$ FT of product = conv of FTs

Next we use the fact that the FT of a sinc is a rect. Specifically:

$\mathcal{F}\left(\frac{\text{Sin}[x/c]}{x/c}\right) = c\,\pi\,\text{UnitBox}[c\,\pi\,x]$

Example:

In[14]:=
```
c = 3;
ft = FourierTransform[Sin[x / c] / (x / c), x, s, FourierParameters → {0, -2 π}] /. s → x;
box = c π UnitBox[c π x];
Plot[{ft, .2 + box}, {x, -.1, .1},
 Exclusions → None,
 PlotRange → All,
 PlotLegends → {"F(sinc)(x/3)", box}]
```

Out[17]=




So our integral becomes

$\frac{1}{2} \left(\mathcal{F}\left(\frac{\text{Sin}[x]}{x}\right) \star \mathcal{F}\left(\frac{\text{Sin}[x/3]}{x/3}\right) \star \ldots\right)(0)$

$= \frac{1}{2} (\pi\,\text{UnitBox}[\pi\,x] \star 3\,\pi\,\text{UnitBox}[3\,\pi\,x] \star \ldots)(0)$

We now study the series of convolutions $\pi\,\text{UnitBox}[\pi\,x] \star 3\pi\,\text{UnitBox}[3\pi\,x] \star \ldots$

The UnitBoxes:

In[18]:=
```
cs = Range[1, 9, 2];
boxes = Table[π c UnitBox[π c x], {c, cs}];
Plot[boxes, {x, -.2, .2},
 Exclusions → None,
 PlotRange → All,
 PlotLegends -> "Expressions"]
```

Out[20]=




They all have unit area:

In[21]:= $\int_{-\infty}^{\infty} \text{boxes}\, dx$

Out[21]= {1, 1, 1, 1, 1}

The boxes’ widths follow a simple pattern as they get smaller:

In[22]:=
```
centerWidth[box_] := ArcLength@ImplicitRegion[Reduce[box == (box /. x → 0)], {x}]
```

In[23]:=
```
centerWidth /@ boxes
```

Out[23]= $\left\{\frac{1}{\pi}, \frac{1}{3\pi}, \frac{1}{5\pi}, \frac{1}{7\pi}, \frac{1}{9\pi}\right\}$

Convolving these boxes together yields a progressively smoother function.

Here is the succession of convolutions

π UnitBox[π x]

π UnitBox[π x] ★ 3 π UnitBox[3 π x]

π UnitBox[π x] ★ 3 π UnitBox[3 π x] ★ 5 π UnitBox[5 π x]

...

In[24]:=
```
convs = FoldList[(Convolve[#1, #2, x, s] /. s → x) &, boxes];
```

In[25]:=
```
Plot[convs, {x, -.3, .3}, PlotRange → All, Exclusions → None, PlotLegends → cs,
 PlotLabel → "Successive box-convolutions shrink the plateau around x=0."]
```

Out[25]=




Zoom in:

In[26]:=
```
Plot[convs, {x, -.3, .3}, PlotRange → {{0, .2}, {3, 3.2}}, Exclusions → None]
```

Out[26]=




At 0, each convolution takes the value $\pi$:

In[27]:=
```
convs /. x → 0
```

Out[27]= $\{\pi, \pi, \pi, \pi, \pi\}$

That’s why the integrals $\int_0^\infty \frac{\text{Sin}[x]}{x} \frac{\text{Sin}[x/3]}{x/3} \frac{\text{Sin}[x/5]}{x/5} \ldots dx$ have value $\pi/2$.

However, successive convolutions’ widths get smaller:

In[28]:=
```
centerWidth /@ convs
% // N
```

Out[28]= $\left\{\frac{1}{\pi}, \frac{2}{3\pi}, \frac{7}{15\pi}, \frac{34}{105\pi}, \frac{67}{315\pi}\right\}$

Out[29]= {0.31831, 0.212207, 0.148545, 0.103072, 0.067704}

In[30]:=
```
ListLinePlot[{cs, %}ᵀ, PlotRange → {Automatic, {0, Automatic}},
 PlotMarkers → {Automatic, 20},
 PlotLabel → "Widths of plateaus of successive convolutions",
 PlotLegends → {"found with symbolic convolutions"}]
```

Out[30]=



Does this sequence stay positive forever, or does it cross zero?

Brute-force find a pattern:

In[31]:=
```
widths = centerWidth /@ convs
```

Out[31]= $\left\{\frac{1}{\pi}, \frac{2}{3\pi}, \frac{7}{15\pi}, \frac{34}{105\pi}, \frac{67}{315\pi}\right\}$

In[32]:=
```
FindSequenceFunction[{cs, widths}ᵀ, x]
```

Out[32]= $\frac{1}{\pi} + \frac{\text{PolyGamma}\left[0, \frac{3}{2}\right] - \text{PolyGamma}\left[0, \frac{1}{2} + \frac{1+x}{2}\right]}{2\pi}$

In[33]:=
```
FullSimplify[%, Assumptions → x > 0]
```

Out[33]= $-\frac{-4 + \text{HarmonicNumber}\left[\frac{x}{2}\right] + \text{Log}[4]}{2\pi}$

In[34]:=
```
widths2 = % /. x → Range[1, 19, 2];
```

In[35]:=
```
ListLinePlot[{{cs, widths}ᵀ, {Range[1, 19, 2], widths2}ᵀ},
 PlotMarkers → {Automatic, 20},
 PlotLegends →
  {"found with symbolic convolutions", "HarmonicNumber seq from MMA"},
 PlotLabel → "Widths of plateaus of successive convolutions"]
```

Out[35]=




So it looks like the $x/15$ term has a “negative” width, indiciating the plateau has been convolved away.

Besides this weird formula with harmonic numbers, there is a simpler way to express the widths of the convolutions. Convolving a box with a smaller box of width $c$ reduces the width of the flat center region by $c$. So the width of successive box-convolutions can be found by successively subtracting box widths from the width of the original box $1/\pi$:

In[36]:=
```
FoldList[Subtract, centerWidth /@ boxes]
```

Out[36]= $\left\{\frac{1}{\pi}, \frac{2}{3\pi}, \frac{7}{15\pi}, \frac{34}{105\pi}, \frac{67}{315\pi}\right\}$

This matches the widths we found with symbolic convolution:

In[37]:=
```
centerWidth /@ convs
```

Out[37]= $\left\{\frac{1}{\pi}, \frac{2}{3\pi}, \frac{7}{15\pi}, \frac{34}{105\pi}, \frac{67}{315\pi}\right\}$

This is good because the closed-form expression of the $x/15$ convolution is too big for Mathematica.

This matches the widths of our symbolic convolutions, as well as our werd formula with harmonic numbers:

In[38]:=
```
widths3 = Block[{cs, boxes, convWidths},
   cs = Range[1, 19, 2];
   boxes = Table[π c UnitBox[π c x], {c, cs}];
   convWidths = FoldList[Subtract, centerWidth /@ boxes]
  ]
```

Out[38]= $\left\{\frac{1}{\pi}, \frac{2}{3\pi}, \frac{7}{15\pi}, \frac{34}{105\pi}, \frac{67}{315\pi}, \frac{422}{3465\pi}, \frac{2021}{45045\pi}, -\frac{982}{45045\pi}, -\frac{61739}{765765\pi}, -\frac{1938806}{14549535\pi}\right\}$

In[39]:=
```
ListLinePlot[
 {{cs, widths}ᵀ, {Range[1, 19, 2], widths2}ᵀ, {Range[1, 19, 2], widths3}ᵀ},
 PlotMarkers → {Automatic, 20},
 PlotLegends → {"found with symbolic convolutions",
   "HarmonicNumber seq from MMA", "Differences in box widths"},
 PlotLabel → "Widths of plateaus of successive convolutions"]
```

Out[39]=




The curve goes negative at x=15, meaning the convolutions have eaten up the entire flat center region at the $x/15$ convolution, so the value of the FT at 0 is less than $\pi/2$. I think it would be difficult to figure out how much less that $\pi/2$ it is; it's probably easiest to just perform the infinite integral.